

## 第4章 回帰分析の諸問題（1）

### 第1節 多重共線性

#### 1. 多重共線性の例

多重共線性の症状について、数値例を用いて確認してみる。次のプログラムを入力し、実行してみよう。（ファイル名は **multico.tsp** としておく）

```
freq n;
smpl 1 5;
?
? 完全な多重共線性
?
load y; 13 4 8 9 6;
load x; 4 1 2 5 3;
load w; 8 2 4 10 6;
print y x w;
corr y x w;
olsq y c x w;
?
? 多重共線性その1
?
load x; 4.01 1 2 5 3;
print y x w;
corr y x w;
olsq y c x w;
?
? 多重共線性その2
?
load x; 4.05 1 2 5 3;
print y x w;
corr y x w;
olsq y c x w;
end;
```

#### 2. VIF値

上の最後の例について VIF 値を計算してみる。`end` 文の前に次のようなコマンドを入力してみよう。

```
?
? VIF 値
?
olsq x c w;
r2x = @rsq;
vifx = 1/(1-r2x);
olsq w c x;
r2w = @rsq;
vifw = 1/(1-r2w);
smpl 1;
print vifx vifw;
```

□演習問題：**ex3-3.tsp** について、同様に VIF 値を求めるコマンドを追加してみよう。そのプログラムは **ex3-3multico.tsp** という名前で保存しておくこと。

## 第2節 系列相関

### 3. Excelをもちいた系列相関の判定と対処法

3章で実習した **ex3-3.xls** の結果について、ダービン・ワトソン比を計算して系列相関の判定し、対処法である、コクラン・オーカット法、一般化最小2乗法を適用してみよう。

#### □ 手順

CE を被説明変数、YD を説明変数とする単回帰分析を分析ツールを用いておこない、結果を **ex3-3sc.xls** の名前で保存する。

標準残差の右側（E列）に残差を1期ずらしてコピーし、e(t-1)を作成する。C列の1期から19期のデータをE列の2期から20期までにコピーする。

F列に  $e(t)-e(t-1)$  を計算する。F26セルに「=C26-E26」と入力し、それをコピーする。

G列に  $((e(t)-e(t-1))^2$  を計算（G26セルなら「=F26^2」）し、その合計を求める。

H列に  $e(t)^2$  を計算（H25セルなら「=C25^2」）し、その合計を求める。

G列の合計を H列の合計で割ったものがダービン・ワトソン比となる。

#### <コクラン・オーカット法>

#### □ 手順

$e(t)$ を  $e(t-1)$ に対して回帰する。分析ツールを用いてよい。ここで「定数に0を使用」にチェックすること。

分析結果のX値1の係数を用いて  $Y_t^*$ ,  $X_t^*$ を計算する。この係数は絶対参照すること。

$Y_t^*$ を  $X_t^*$ に対して回帰する。これがコクラン・オーカット法によって求めた係数推定値である。 $b$  の推定値は分析結果そのままであるが、 $a$  の推定値はこの結果を  $1 - b$  で割ったものになる。また、この結果について残差プロットやダービン・ワトソン比によって系列相関への対処がなされたことを確認すること。

#### <一般化最小2乗法>

#### □ 手順

コクラン・オーカット法と同様に  $e(t)$ を  $e(t-1)$ に対して回帰する。コクラン・オーカット法の計算のすでに計算してあるので、このステップは省略できる。

分析結果のX値1の係数を用いて  $Y_t^*$ ,  $X_t^*$ ,  $Z_t^*$ を計算する。

$Y_t^*$ を  $X_t^*$ ,  $Z_t^*$ に対して回帰する。「定数に0を使用」にチェックすること。このようにしても求められたものが一般化最小2乗法による係数推定値である。

#### 4. TSPをもちいた系列相関の判定と対処法

Excel では、系列相関の判定や対処法に若干の工夫が必要であった。TSP ではそれが容易におこなうことができる。次のようなプログラムを作成し、**ex3-3sc.tsp** という名前で保存せよ。

```
freq a;
smpl 1983 2002;
read(file='h:ex3-3.csv') Date ce yd yd1;
?
? 単回帰と残差プロット
?
olsq ce c yd;
ste=@res/@s;
graph date ste;
?
? の導出とコクラン・オーカット法
?
olsq @res @res(-1);
ce2=ce-@coef*ce(-1);
yd2=yd-@coef*yd(-1);
olsq ce2 c yd2;
?
? AR1 コマンドの利用
?
ar1(method=corc) ce c yd;
ar1(method=ML) ce c yd;
end;
```

このプログラムでは、コクラン・オーカット法を Excel で実習したのと同様の方法でも計算しているが、AR1 コマンドのみでも十分である。

### 第3節 系列相関

#### 5. 不均一分散の対処法

**hetero.xls** は 27 の会社の労働者数と管理者数のデータである。このデータについて不均一分散があるかどうかを調べてみる。

残差プロットと LM het. test の値から不均一分散が示唆される。

対処法として、 $Y/X$  を被説明変数、 $1/X$  を説明変数とする回帰分析を行うことがある。これによって不均一分散の問題は解決できたといえよう。

#### プログラム

```
freq n;
read(file='h:hetero.xls');
print y x;
graph x y;
olsq y c x;
ste=@res/@s;
graph x ste;
yy=y/x;
xx=1/x;
olsq yy c xx;
ste=@res/@s;
graph xx ste;
end;
```